Development of the Theory of Continuous Lattices in Mizar

Grzegorz Bancerek
Institute of Computer Science
University of Białystok
email: bancerek@mizar.org

July 1, 2000

Abstract. *This paper reports on* Mizar *formalization of the theory of continuous lattices included in the* A Compendium of Continuous Lattices, *[7].* Mizar *formalization means a formalization of theorems, definitions, and proofs in the* Mizar *language such that it is accepted by the* Mizar *system. This effort was originally motivated by the question whether the* Mizar *system is sufficiently developed as to allow expressing advanced mathematics. The current state of the formalization, which includes 49* Mizar *articles written by 14 authors, suggests that the answer is positive. The work of the team of authors in cooperation with the Library Committee*[1] *and system designers resulted in improvements of the system towards a more convenient technology for doing mechanically checked mathematics. It revealed, also, that the substantial element of the convenience is the incorporation of computer algebra into* Mizar *system.*

1 Introduction

To formalize means to investigate some mathematical theory rigorously, obeying fixed rules of formulating, defining, proving, and reasoning. Mizar

[1]The Library Committee of the Association of Mizar Users maintains the Mizar Mathematical Library and coordinates activities concerning improvement of the library.

formalization admits some variety of expression but the required rigor assures that the result of the formalization, that is a text in the context of the MIZAR system, has unique meaning. When formalizing a theory we introduce definitions, lemmas, and theorems with the hope that they will be useful for future developments. This is the essential idea behind developing the MIZAR data base.

MIZAR[2] has been designed by Andrzej Trybulec and developed by a team under his leadership. The system includes a language, software tools, a library, and a hyperlinked journal.

The MIZAR language is an attempt to approximate mathematical vernacular in a formal language. Reserved words form a subset of English words which are used in regular mathematical papers with the same meaning. The logic of MIZAR is classical, the proofs are written in the Fitch-Jaśkowski style, see [9]. Definitions allow to introduce type, term, and formula constructors and require proving of correctness conditions. A proof consists of a sequence of steps, each step justified by facts proved in earlier steps, lemmas, theorems and/or schemes. Schemes are second order theorems which may be used to formulate e.g. induction. Multi prefixed structures allow to introduce algebraic concepts, for example topological groups which are both groups and topological spaces. More detailed description of MIZAR system can be found in [6], [18], and, also, in [13].

MIZAR software includes tools supporting some typical tasks when doing mathematics:

- development and management of knowledge base,
- verification of logical correctness,
- elements of generalization, simplification, readability enhancement,
- presentation using TEX and HTML.

The Mizar Mathematical Library (MML) is a collection of texts written in MIZAR language called *Mizar articles.* The MML is based on the Tarski-Grothendieck set theory. As of March 2000 there were 633 articles collected. They included 29,514 theorems, 5,389 definitions and redefinitions, and 317,427 references to external theorems (i.e. in other articles).

Mizar articles are automatically translated into English and published in *Formalized Mathematics.* The electronic version, *Journal of Formalized Mathematics*, `http://www.mizar.org/JFM/` includes hyper-links to definitions which substantially help in using the MML. More details may be found on MIZAR web pages at `http://www.mizar.org/`.

[2]Mizar is a star; ζ in Ursa Major.

MIZAR might be considered as an Esperanto for mathematics and there are some similarities between both languages. Esperanto was developed in Białystok by L. Zamenhoff and MIZAR is being developed in Białystok. Esperanto is an artificial international language with words taken from several national languages and uses a quite regular grammar. MIZAR may be considered as an attempt to standardize the language of mathematics. There is a lot of translations of books into Esperanto and it is possible to learn the language by reading those translations. There is a large collection of MIZAR articles and it is possible to learn MIZAR (and mathematics) by reading them.

2 Goal and Motivations

At the QED[3] Workshop II, Warsaw 1995, the following question was raised:

> Can we do formalization of advanced mathematics like this included in regular mathematical books in the current proof-checking systems?

In trying to answer this question we have decided to put MIZAR into a serious test. We have chosen *A Compendium of Continuous Lattices* [7] to be formalized in its entirety. The theory of continuous lattices presented in [7] is mathematically advanced. It involves a variety of areas of mathematics: computation, topology, analysis, algebra, category theory, and logic. Also, it is a relatively recent and a well-established field. The choice turned out to be a lucky one. The compendium is very rigorous which made the formalization comparatively easy; also, some initial fragments of the theory of lattices had been already developed in MIZAR.

In the past, there were some attempts to formalize entire mathematical books in computerized proof-checking systems. In the 1970's, Jutting [17] formalized Landau's *Grundlagen* [12] in AUTOMATH. Another attempt was the formalization of 2 chapters of *Theoretical Arithmetic* by Grzegorczyk, [8], in the 1980's. It was done by A. Trybulec's team in MIZAR 2, which was not equipped with the library. In MIZAR 2, each text was processed separately from other texts. All background knowledge needed to write a text was put without proofs in a preliminary section.

In 1989 we started to collect all MIZAR texts and on this basis develop and maintain the Mizar Mathematical Library. Each new MIZAR article can be submitted to the MML if it is accepted by the MIZAR verifier and

[3] http://www-unix.mcs.anl.gov/qed/

refers only to articles already included in the MML. At the start, the basis of the MML was formed by two axiomatic articles:

- *Tarski Grothendieck set theory* [15],
- *Built-in concepts* [14], including strong arithmetic of real numbers.

The latter became a regular MIZAR article in 1998 when the construction of real numbers was completed.

The experiment with formalization of an entire book has many aspects and we expected to get answers to the following:

- Is the MIZAR language sufficiently expressive to formulate definitions, theorems, and proofs contained in [7]?
- Is MML rich enough to even start the formalization? Did MML cover the knowledge assumed in the compendium as background?
- Can the different concepts already defined in independent articles in MML be used together?
- Is the MIZAR software capable of handling this amount of material?

We hoped that running such an experiment, irrespective of the answers to the above questions, would lead to an improvement of MIZAR.

3 Teamwork

The work performed on the formalization is a result of a team effort by the researchers and students of the Institute of Mathematics and Institute of Computer Science, University of Białystok. MIZAR articles written in the project have been authored by:

Czesław Byliński, Adam Grabowski, Ewa Grądzka, Jarosław Gryko, Artur Korniłowicz, Beata Madras, Agnieszka J. Marasik, Robert Milewski, Adam Naumowicz, Piotr Rudnicki (University of Alberta, Canada), Bartłomiej M. Skorulski, Andrzej Trybulec, Mariusz Żynel, and Grzegorz Bancerek.

In the summer of 1995, we started a seminar devoted to the theory of continuous lattices following [7] and [10]. In the spring of 1996, the final decision on formalization of [7] was made. Parts of the first two chapters, *O. A Primer of Continuous Lattices* and *I. Lattice Theory of Continuous Lattices*, were assigned to individual team members for formalization. We adopted the following rules:

- formalization is divided into two series of Mizar articles with the identifiers[4]:
  - YELLOW - articles bridging the MML and the knowledge assumed in the compendium,
  - WAYBEL[5] - articles formalizing the main course of the compendium,
- no formalization of examples unless necessary,
- the formalization is as close to [7] as possible but taking into account some Mizar peculiarities such as built-in concepts and mechanisms, possibility of automatic generalization, reuse of the MML, etc.
- the formalization should be more general than the theory in the compendium, e.g. we follow hints at generalization included in exercises (see 1.26 on page 52 of [7] and compare pages 38–42 with [4]).

Because of the number of people involved, the work was organized in a different way than the usual sequential contributions of articles to MML. Usually, an author writes an article and the article is not available to other authors until it is submitted to the MML. We wanted to formalize different parts of the book simultaneously as sequential development would be too slow. We decided to maintain a local library of YELLOW and WAYBEL series with completed and non-completed articles. This allowed for some parallelism in writing articles. The articles from local library were tested by later ones that used them and if there was a need they were revised. After some time they were presented on a seminar to discus possible generalization and, finally, submitted to the MML.

The size of the YELLOW series (17 articles of 49) indicates that MML was almost ready for the formalization. However, the following topics had to be developed:

- upper and lower bounds, suprema and infima (YELLOW_0),
- poset under inclusion (YELLOW_1),
- lattice of ideals (YELLOW_2),
- complete lattices (YELLOW_0 and YELLOW_2),

[4]Each Mizar article, besides regular title, has an identifier which is used when referring to it.

[5]The *way below* relation is the key concept in continuous lattices - it is used to characterize continuous lattices.

- Cartesian product of posets and lattices (YELLOW_1, YELLOW_3, and YELLOW10),
- lattice operations on subsets of a poset (YELLOW_4),
- Boolean lattices (YELLOW_2 and YELLOW_5),
- duality in lattices (YELLOW_7),
- modular and distributive lattices (YELLOW11),
- Moore-Smith convergence (YELLOW_6),
- Baire spaces and sober spaces (YELLOW_8),
- bases of topologies (YELLOW_9, YELLOW13, and YELLOW15),
- refinements of topologies (YELLOW_9),
- Hausdorff spaces (YELLOW12),
- product of topological spaces (YELLOW14),
- topological and poset retracts (YELLOW16).

The formalization was a stress test for the MIZAR software. It detected some errors and forced adjusting a number of quantitative parameters. The formalization would not be possible without cooperation with system designers and the Library Committee in improvement of software and a number of revisions to the MML.

4 Defining Lattices in MIZAR

There are at least two approaches to lattices in mathematics. According to the first, a lattice is an algebra with two binary operations $\sqcup$ and $\sqcap$ which satisfy the conditions of idempotency, associativity, commutativity, and absorption. According to the second, a lattice is a partially ordered set (poset) with suprema and infima for non empty finite subsets. Both approaches were already present in the MML and the correspondence between them was proved [19, 16, 1]. The second approach gives wider usage and is easier to generalize (e.g. by weakening the condition of partial ordering). This approach was chosen and the first revision of the MML consisted in generalization of posets and some lattice-theoretical concepts.

`RelStr` is the base structure of quasi ordered sets, posets, semilattices, and lattices and was introduced in [16] as follows:

```
definition
  struct (1-sorted) RelStr (#
    carrier     -> set,
    InternalRel -> Relation of the carrier
  #);
end;
```

If R is RelStr then R is a structure with at least 2 fields: carrier and InternatRel. A structure S can be a RelStr and may have more fields when its type is derived from RelStr. This is not the case when S is strict RelStr. The definition of the attribute strict is generated automatically by each structure definition.

The concept of a poset was introduced as follows:

```
definition
  mode Poset is reflexive transitive antisymmetric RelStr;
end;
```

The attributes reflexive, transitive, and antisymmetric and the following existential cluster registration were introduced earlier.

```
definition
  cluster non empty reflexive transitive antisymmetric strict RelStr;
  existence
   proof
    :: Demonstration that such an object exists
   end;
end;
```

(Two colons :: start a comment which ends at the end of the line.)

The above cluster assures existence of a RelStr type objects with any subset of the listed attributes.

For convenience and to be closer to usual notation the following definition was introduced.

```
definition
  let R be RelStr;
  let x, y be Element of the carrier of R;
  pred x <= y means
:: ORDERS_1:def 9
   [x,y] in the InternalRel of R;
  synonym y >= x;
end;
```

The characterizations of reflexivity, transitivity, and antisymmetry were given in [2] as redefinitions:

```
definition
  let A be non empty RelStr;
 redefine
  attr A is reflexive means
:: YELLOW_0:def 1
    for x being Element of A holds x <= x;
   compatibility proof .... end;
end;

definition
  let A be RelStr;
 redefine
  attr A is transitive means
:: YELLOW_0:def 2
    for x,y,z being Element of A st x <= y & y <= z holds x <= z;
   compatibility proof .... end;
  attr A is antisymmetric means
:: YELLOW_0:def 3
    for x,y being Element of A st x <= y & y <= x holds x = y;
   compatibility proof .... end;
end;
```

The concept of lattice was introduced by definitions:

```
definition
  let R be RelStr;
  attr R is with_join means
:: LATTICE3:def 10
    for x,y being Element of R
     ex z being Element of R st x <= z & y <= z &
      for z' being Element of R st x <= z' & y <= z' holds z <= z';
  attr R is with_meet means
:: LATTICE3:def 11
    for x,y being Element of R
     ex z being Element of R st z <= x & z <= y &
      for z' being Element of R st z' <= x & z' <= y holds z' <= z;
end;

:: WAYBEL_0
definition
 mode Semilattice is with_meet Poset;
 mode sup-Semilattice is with_join Poset;
 mode LATTICE is with_join with_meet Poset;
end;
```

5 Continuous Lattices

The concept of directed sets was changed in MIZAR formalization. A directed set is non empty in mathlore and in the compendium. However, it happens often that we need a set which is directed or empty. MIZAR does not allow to write a type as *directed or empty set* and we decided to formalize the concept as follows:

```
definition
  let L be RelStr;
  let X be Subset of L;
  attr X is directed means
:: WAYBEL_0:def 1     :: CCL, Definition 1.1, p. 2
   for x,y being Element of L st x in X & y in X
      ex z being Element of L st z in X & x <= z & y <= z;
  attr X is filtered means
:: WAYBEL_0:def 2     :: CCL, Definition 1.1, p. 2
   for x,y being Element of L st x in X & y in X
      ex z being Element of L st z in X & z <= x & z <= y;
end;
```

The theorem explaining correspondence to usual meaning has been proved also.

```
theorem :: WAYBEL_0:1
 for L being non empty transitive RelStr, X being Subset of L holds
   X is non empty directed iff
    for Y being finite Subset of X
     ex x being Element of L st x in X & x is_>=_than Y
  proof .... end;
```

The concept of completeness presented in [7] depends on a context. A complete poset, complete semilattice, and complete lattice satisfy different conditions. In MIZAR we introduced attributes

- `up-complete` as completeness with respect to directed sups,
- `inf-complete` as completeness with respect to non empty infs,
- `complete` as completeness with respect to all sups.

Then, in MIZAR notation

Compendium	MML
a *complete poset*	`up-complete Poset`
a *complete semilattice*	`inf-complete up-complete Semilattice`
a *complete lattice*	`complete LATTICE`.

The fact that a complete lattice is a complete poset and a complete semi-lattice is expressed in MIZAR (see [3]) by conditional cluster registration:

```
definition
  cluster complete -> up-complete inf-complete (non empty reflexive RelStr);
  coherence
   proof
     let R be non empty reflexive RelStr;
     assume R is complete;
     ....
     thus R is up-complete by ...
     ....
     thus R is inf-complete by ...
   end;
end;
```

The conditional registration is used automatically by MIZAR. Attributes `up-complete` and `inf-complete` are added to a type when it widens to `non empty reflexive RelStr` and already includes attribute `complete`. The concept of continuous lattices presented in the compendium depends on context. We decided to formalize it in as general way as possible because all meanings of it may be expressed by the basic `continuous` attribute and some extra conditions of completeness.

```
definition
  let L be non empty reflexive RelStr;
  attr L is continuous means
:: WAYBEL_3:def 6
   (for x being Element of L holds waybelow x is non empty directed) &
   L is up-complete satisfying_axiom_of_approximation;
end;
```

The attribute `satisfying_axiom_of_approximation` is introduced as follows.

```
definition
  let L be non empty reflexive RelStr;
  attr L is satisfying_axiom_of_approximation means
:: WAYBEL_3:def 5
    for x being Element of L holds x = sup waybelow x;
end;
```

The `sup` is the supremum operation, see [2]. The `waybelow x` is a set of all elements of `L` which are way below `x`, see [4].

The MIZAR notation for continuous posets:

Compendium	MML
a *continuous poset*	`continuous up-complete Poset`
a *continuous semilattice*	`continuous up-complete Semilattice`
a *complete-continuous semilattice*	`continuous inf-complete up-complete Semilattice`
a *continuous lattice*	`continuous complete lattice`

As the test of the correctness of the introduced concepts, the correspondence between locally compact topological spaces and continuous lattices has been proved. This correspondence is expressed by two theorems:

```
theorem :: WAYBEL_3:42
 for T being non empty TopSpace
     st T is_T3 & InclPoset(the topology of T) is continuous
  holds T is locally-compact;

theorem :: WAYBEL_3:43
 for T being non empty TopSpace st T is locally-compact
  holds InclPoset(the topology of T) is continuous;
```

The `InclPoset(the topology of T)` is the poset of open sets from space `T` ordered by inclusion.

6 Mixing Order and Topology

Topologies on posets induced by the ordering and, conversely, partial orders on topological spaces generated by topology are investigated in the theory of continuous lattices. For example, *Scott topology* introduced in the compendium is the family of sets which are inaccessible by directed sups. *Lawson topology* and *lower topology* are another example of such topologies. *Lawson topology* is the common refinement of *Scott* and *lower topologies* and *lower topology* is generated by complements of principal filters as subbasic open sets.

When investigating such topologies we need to use both theories: posets and topological spaces. In the case of *Lawson topology* we have in the same time three topologies and a poset. The solution from the compendium consists in introducing new notation like *Scott open*, *Scott closed*, *Scott neighbourhood*, etc. It is possible to do the same in Mizar but such notation causes substantial technical difficulties with the use of general topology developed in the MML. Besides, such notation is not consequently applied in the compendium.

The problem was solved by multi prefixed structure definition in [11] and by mode definition in [5].

```
:: WAYBEL_9
struct(TopStruct, RelStr)
 TopRelStr (# carrier -> set,
              InternalRel -> (Relation of the carrier),
              topology -> Subset-Family of the carrier #);

definition
 let R be RelStr;
 mode TopAugmentation of R -> TopRelStr means
:: YELLOW_9:def 4
  the RelStr of it = the RelStr of R;
  existence proof .... end;
end;
```

`TopStruct` has two fields: `carrier` and `topology` and is the base structure of topological spaces. The structure `TopRelStr` is both the structure `TopStruct` and the structure `RelStr`. We may apply to it attributes defined for posets and attributes defined for topological spaces as well. If `X` is `TopRelStr`, then `the RelStr of X` will be `strict RelStr` and, moreover,

```
the RelStr of X = RelStr(# the carrier of X, the InternalRel of X #)
```

(analogically, for `TopStruct`).

```
:: WAYBEL_9
definition
 mode TopLattice is with_join with_meet reflexive transitive
        antisymmetric TopSpace-like TopRelStr;
end;
```

As an illustration of applied convention, let us compare the proposition 1.6 from the compendium, page 144, and corresponding MIZAR theorems.

> 1.6. PROPOSITION. *Let* L *be a complete lattice.*
> (i) *An upper set U is Lawson open iff it is Scott open*;
> (ii) *A lower set is Lawson-closed iff it is closed under sups of directed sets.*

```
theorem :: WAYBEL19:41
:: 1.6. PROPOSITION (i), p. 144
 for S being Scott complete TopLattice
 for T being Lawson correct TopAugmentation of S
 for A being upper Subset of T st A is open
 for C being Subset of S st C = A holds C is open;

theorem :: WAYBEL19:42
```

```
:: 1.6. PROPOSITION (ii), p. 144
 for T being Lawson (complete TopLattice)
 for A being lower Subset of T holds
   A is closed iff A is closed_under_directed_sups;
```

The implication from right to left in point (i) is proved in more general case:

```
theorem :: WAYBEL19:37
 for S being Scott complete TopLattice
 for T being Lawson correct TopAugmentation of S
 for A being Subset of S st A is open
 for C being Subset of T st C = A holds C is open;
```

7 Some statistics

The project started in 1996. The compendium contains 334 pages and the theory formalized by the end of February 2000 covers about 180 pages of it (about 54% without taking into account the articles currently under development).

The following summarizes the number of articles from this project submitted to MML:

year	1996	1997	1998	1999	2000	1996–2000
YELLOW	8	1	5	3	0	17
WAYBEL	10	6	8	4	4	32
All	18	7	13	7	4	49
Y÷All %	44%	14%	38%	42%	0%	35%

The last line gives percentage of YELLOW series. This percentage is less than we expected.

	MML	WAYBEL	YELLOW	W&Y	Percentage
Articles	633	32	17	49	7.74%
Theorems	29,514	1,391	834	2225	7.54%
ave per art	46.6	43.5	49.1	45.4	–
Definitions	5,389	246	105	351	6.51%
ave per art	8.5	7.7	6.2	7.2	–
Size (kB)	46,966	2,867	1,273	4,140	8.82%
ave per art	74.2	89.6	74.9	84.5	–

The last column gives percentage of this project in the entire MML. Average numbers of theorems, definitions, and kilobytes show that the project is close to the MML average. Note the smaller average number of definitions which may indicate that the theory is explored more intensively.

The interaction between the project and the rest of the MML may be measured by the number of references between them. Each reference to the theorem coming from another article is called an external reference.

External references	Count	Percentage
All	317427	100.00%
All to Y&W	11677	3.68%
Outside of Y&W to Y&W	349	0.11%
In Y&W	26747	100.00%
In Y&W to Y&W	11328	42.35%

57.65% of all external references from the YELLOW and WAYBEL articles is to the rest of the MML. This indicates that the MML contained a substantial quantity of definitions and facts needed for our project.

There is, unfortunatly, no statistics concerning the quantity of work needed to formalize this material. However, we may state that it vary on authors and WAYBEL series needed much more work per line than YELLOW series.

8 Conclusions

Our main conclusion is that the Mizar system seems satisfactory to formalize advanced mathematics.

The second conclusion is that the MML was satisfactorily rich to start formalization of the compendium. The YELLOW series constitutes only 35% of the whole project.

Formalization in Mizar is still not as simple as doing mathematics traditionally. It should be improved in near future. Now, however, there are some gains. The results are mechanically checked. There is an automatic access to the knowledge stored and the net of concepts is explicit. (This helped very much for new authors to start.) The information may be mechanically explored: changed, generalized, and edited. Reorganization of a machine readable mathematical text is much easier than reorganization of such a text written on paper. (Such reorganizations were quite often required in our project.)

The work done in this project resulted in numerous improvements of the Mizar system and, also, it revealed a number of issues that are investigated:

- tools for searching MML: semantic searching which can distinguish homonyms, glue synonyms, and recognize hidden arguments,
- proof assistance based on the exploration of existing proofs in the MML and computer algebra,
- reorganization of the MML: revisions of existing articles and "online revision" mechanism available by environment directives,
- development of Mizar language to improve the convenience of formulation of definitions and proofs, the length of proofs, and the flexibility of type structure: type modifier, attributes with explicit arguments, new realization of structure types.

References

[1] Grzegorz Bancerek. Complete lattices. *Formalized Mathematics*, 2(**5**):719–725, 1991. MML: LATTICE3.

[2] Grzegorz Bancerek. Bounds in posets and relational substructures. *Formalized Mathematics*, 6(**1**):81–91, 1997. MML: YELLOW_0.

[3] Grzegorz Bancerek. Directed sets, nets, ideals, filters, and maps. *Formalized Mathematics*, 6(**1**):93–107, 1997. MML: WAYBEL_0.

[4] Grzegorz Bancerek. The "way-below" relation. *Formalized Mathematics*, 6(**1**):169–176, 1997. MML: WAYBEL_3.

[5] Grzegorz Bancerek. Bases and refinements of topologies. *Formalized Mathematics*, 7(**1**):35–43, 1998. MML: YELLOW_9.

[6] Ewa Bonarska. *An Introduction to PC Mizar*. Fondation Philippe le Hodey, Brussels, 1990. Revised version, 2000: http://merak.uwb.edu.pl/~bancerek/introduction/.

[7] G. Gierz, K. H. Hofmann, K. Keimel, J. D. Lawson, M. W. Mislove, and D. S. Scott. *A Compendium of Continuous Lattices*. Springer-Verlag, Gerlin, Heidelberg, 1980.

[8] Andrzej Grzegorczyk. *Zarys arytmetyki teoretycznej*. Biblioteka Matematyczna. PWN, Warszawa, 1971.

[9] Stanisław Jaśkowski. *On the Rules of Supposition in Formal Logic*. Studia Logica. Warsaw University, 1934. Reprinted in S. McCall, Polish Logic in 1920–1939, Clarendon Press, Oxford.

[10] Peter T. Johnstone. *Stone Spaces.* Cambridge University Press, Cambridge, London, New York, 1982.

[11] Artur Korniłowicz. On the topological properties of meet-continuous lattices. *Formalized Mathematics*, 6(**2**):269–277, 1997. MML: WAYBEL_9.

[12] E. G. H. Landau. *Grundlagen der Analysis.* Akademische Verlag, Leipzig, 1930.

[13] Piotr Rudnicki and Andrzej Trybulec. On equivalents of well-foundedness. *Journal of Automated Reasoning*, 23(3-4):197–234, 1999.

[14] Andrzej Trybulec. Built-in concepts. *Formalized Mathematics*, 1(**1**):13–15, 1990. MML: AXIOMS.

[15] Andrzej Trybulec. Tarski Grothendieck set theory. *Formalized Mathematics*, 1(**1**):9–11, 1990. MML: TARSKI.

[16] Wojciech A. Trybulec. Partially ordered sets. *Formalized Mathematics*, 1(**2**):313–319, 1990. MML: ORDERS_1.

[17] L. S. van Benthem Jutting. Checking Landau's "Grundlagen" in the Automath system, 1977. PhD thesis.

[18] Freek Wiedijk. *Mizar: An Impression*, 1999. http://www.cs.kun.nl/~freek/mizar/.

[19] Stanisław Żukowski. Introduction to lattice theory. *Formalized Mathematics*, 1(**1**):215–222, 1990. MML: LATTICES.